



**SB\_5.1: Summenverstärker mit bewerteten Eingängen**




Bild SB\_5.1\_1: Schaltung zum Simulationsbeispiel 5.1

\*\*\* Netzliste SB\_5.1 \*\*\*

```
V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Ue       3  0  100mV
X_N1       1  2  V+ V-  out uA741
R_RN       2  out 40k
R_R1       3  2  20k
R_R2       3  2  10k
R_R3       3  2  5k
R_R4       3  2  2.5k
R_Rp       0  1  1.1k
R_Ra       0  out 10MEG
.TEMP      27          .PROBE          .END
```

Abschn.: **5.10**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

DC-Main-Sweep

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** unter: *View* → *Output File*

(Arbeitsoberfläche SCHEMATICS)

*Analysis* → *Examine Output*

(PROBE-Fenster)

\*\*\*\* SMALL SIGNAL BIAS SOLUTION TEMPERATURE = 27.000 DEG C

\*\*\*\*\*

Ende dieses Beispiels



**SB\_5.2: Differenzverstärker**




Bild SB\_5.2\_1: Schaltung zum Simulationsbeispiel 5.2

\*\*\* Netzliste SB\_5.2 \*\*\*

```
V_Ue1      4  0  SIN 0  1V 1kHz  0  0  0
V_Ue2      2  0  SIN 1V 2V 2kHz  0  0  0
V_UB1      V+  0  15V
V_UB2      0  V- 15V
X_N1       3  1  V+  V-  out uA741
R_R2       2  1  10k
R_RN       1  out 30k
R_R1       4  3  10k
R_RP       0  3  30k
.TRAN      0  1.5ms  0  1u
.TEMP      27
.PROBE
.END
```

Abschn.: **5.10**

DC-Analyse  
AC-Analyse  
**Tran.-Analyse**  
DC-Main-Sweep  
DC-Nested-Sweep  
AC-Sweep  
Param.-Sweep  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (x)

OK



**SB\_5.3: Differenzierer**




Bild SB\_5.3\_1: Schaltung zum Simulationsbeispiel 5.3

\*\*\* Netzliste SB\_5.3 \*\*\*

```
V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Ue       in  0  PULSE 0V 1V 0 1ms 0.5ms 1ms 3ms
X_N1       0  1  V+ V- out uA741
R_RN       1  out 20k
R_R1       in  2  125
C_C1       2  1  50n
R_Ra       0  out 10MEG
.TRAN      0  3ms 0  1u
.TEMP      27
.PROBE
.END
```

Abschn.: **5.10**

DC-Analyse  
**AC-Analyse**  
**Tran.-Analyse**

DC-Main-Sweep  
DC-Nested-Sweep  
**AC-Sweep**  
**Param.-Sweep**  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: **V (out)**

**OK**



**SB\_5.4: Integrierer mit Anfangsbedingungen**




Bild SB\_5.4\_1: Schaltung zum Simulationsbeispiel 5.4

\*\*\* Netzliste SB\_5.4 \*\*\*

```
V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Ue       in  0  PULSE 0V 2V 100us 0 0 200us 500us
V_UH       2  0  2V
X_N1       0  3  V+  V-  out  LF411
X_Sw1      3  1  4  Sw_perChange PARAMS: t_D=100us
+ t_2=300us t_periode=600us t_switch=1us R_low=1m R_high=1g
C_CN       3  out 50n IC=0
R_RN       1  out 300
R_R2       2  1  300
R_R1       in  4  2k
R_Ra       0  out 10MEG
.TRAN      0  1ms 0  1u
.TEMP      27
.PROBE
.END
```

Abschn.: **5.10**

DC-Analyse  
AC-Analyse  
**Tran.-Analyse**

DC-Main-Sweep  
DC-Nested-Sweep  
AC-Sweep  
Param.-Sweep  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (out)

OK

Ende dieses Beispiels



**SB\_5.5: Fensterkomparator**




Bild SB\_5.5\_1: Schaltung zum Simulationsbeispiel 5.5

\*\*\* Netzliste SB\_5.5 \*\*\*

```
V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Ue       in  0  DC 0  AC 0
+PWL 1m 0 2m 2 3m -2 4m 0 5m 4 6m 0.5 7m -3 8m 1.5 9m -2 10m 2
V_Uref1    5  0  1V
V_Uref2    1  0 -1V
X_N1       5  in  V+ V- 6  uA741
X_N2       in  1  V+ V- 2  uA741
D_V1       out 2  D1N4148
D_V2       out 6  D1N4148
R_R1       out V+ 10k
R_Ra       out 0  10MEG
.TRAN      0  10ms  0  1u
.TEMP      27                      .PROBE                      .END
```

Abschn.: **5.10**

DC-Analyse  
AC-Analyse  
**Tran.-Analyse**

DC-Main-Sweep  
DC-Nested-Sweep  
AC-Sweep  
Param.-Sweep  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace-Expression-Zeile: **V (out)**

**OK**

Ende dieses Beispiels



**SB\_5.6: A/D-Umsetzung nach dem Parallelverfahren**




Bild SB\_5.6\_1: Schaltung eines Komparators im Simulationsbeispiel 5.6

\*\*\* Netzliste (1) SB\_5.6 \*\*\*

```
V_UB1      V+  0   15V
V_UB2      0   V-  15V
V_Ue       P   0   1V
X_N1       P   N   V+  V-  out LF411
R_R1       V+  N   14k
R_R2       N   V-  16k
R_Ra       0   out 10MEG
.DC        LIN   V_Ue  0   2   1m
.TEMP      27
.PROBE
.END
```

Abschn.: **5.10**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (out)

OK




Bild SB\_5.6\_2: Schaltung des A/D-Umsetzers im Simulationsbeispiel 5.6

\*\*\* Netzliste (2) SB\_5.6 \*\*\*

```

V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Ue       in  0  10mV
V_Uref     M8  0  60mV
X_N1       in  M1 V+ V- out1   LF411
X_N4       in  M4 V+ V- out4   LF411
X_N7       in  M7 V+ V- out7   LF411
R_R0       0   M1 5k
R_R1       M1  M4 10k
R_R4       M4  M7 10k
R_R7       M7  M8 5k
R_Ra1      0   out1 10MEG
R_Ra4      0   out4 10MEG
R_Ra7      0   out7 10MEG
.DC        LIN   V_Ue 0 140mV 10u
.TEMP      27           .PROBE           .END
    
```

Abschn.: **5.10**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (outx)

OK



**SB\_5.7: Spannungsgesteuerte Stromquelle**




Bild SB\_5.7\_1: Schaltung zum Simulationsbeispiel 5.7

\*\*\* Netzliste SB\_5.7 \*\*\*

```
.V_UB1      V+  0  15V
.V_UB2      0  V- 15V
.V_Ue       4  0  1V
.X_N1       3  1  V+ V- 2  uA741
.R_RNx      1  2  3k
.R_RPx      3  2  3k
.R_RN       0  1  1k
.R_RP       4  3  1k
.R_RL       3  0  {Rvar}
.PARAM      Rvar=1k
.DC         LIN  V_Ue  -1.5 1.5 1m
.STEP       PARAM Rvar  LIST  2k 3k 4k 5k 10k
.TEMP       27
.PROBE
.END
```

Abschn.: **5.10**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

DC-Nested-Sweep

AC-Sweep

**Param.-Sweep**

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (2)

OK

Ende dieses Beispiels





**SB\_5.8: Tiefpass mit Elektrometerverstärker**




Bild SB\_5.8\_1: Schaltung zum Simulationsbeispiel 5.8

\*\*\* Netzliste SB\_5.8 \*\*\*

```
V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Ue       3  0  AC 1V  SIN 0 1V 1kHz 0 0 0
X_N1A      1  2  V+ V-  out LM324
R_R1       out 2  2k
R_R2       2  0  1k
R_R3       3  1  10k
R_R4       0  1  {Rc}
C_C4       0  1  3.2n
.PARAM     Rc=1MEG
.AC        DEC 1000 100 1MEG
.STEP     PARAM Rc LIST 5k 10k 100k 1MEG
.TEMP     27
.PROBE
.END
```

Abschn.: **5.10**

DC-Analyse

**AC-Analyse**

Tran.-Analyse

DC-Main-Sweep

DC-Nested-Sweep

**AC-Sweep**

**Param.-Sweep**

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (out)

OK

Ende dieses Beispiels



**SB\_5.9: Tiefpass mit Zweifachgegenkopplung**




Bild SB\_5.9\_1: Schaltung zum Simulationsbeispiel 5.9

\*\*\* Netzliste SB\_5.9 \*\*\*

```
V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Uq       in 0  AC 1V SIN 0 1V 1k 0 0 0
X_N1       0  1  V+ V- out uA741
R_R1       in 2  2.8k
R_R2       2  out 14k
R_R3       2  1  2.8k
C_C1       0  2  {25n/a}
C_C2       1  out {1n*a}
R_Ra       0  out 1MEG
.PARAM     a=1
.AC        DEC 1000 100 100k
.STEP      PARAM a LIST 0.6 1 1.414 1.8
.TEMP      27
.PROBE
.END
```

Abschn.: **5.10**

DC-Analyse  
**AC-Analyse**  
Tran.-Analyse  
  
DC-Main-Sweep  
DC-Nested-Sweep  
**AC-Sweep**  
**Param.-Sweep**  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (out)

OK



**SB\_5.10: Bandpass mit Zweifachgegenkopplung**




Bild SB\_5.10\_1: Schaltung (1) zum Simulationsbeispiel 5.10

\*\*\* Netzliste (1) SB\_5.10 \*\*\*

```
V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Uq       in  0  AC 100mV SIN 0 1V 1k 0 0 0
X_N1       0  1  V+ V- out uA741
R_R1       in  2  10k
R_R2       1  out 300k
R_R3       0  2  2.4k
C_C2       out 2  2.2n
C_C3       1  2  2.2n
R_Ra       0  out 1MEG
.AC        DEC  1000  100 100k
.TEMP      27
.PROBE
.END
```

Abschn.: **5.10**

DC-Analyse

**AC-Analyse**

Tran.-Analyse

DC-Main-Sweep

DC-Nested-Sweep

**AC-Sweep**

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (out)

OK




Bild SB\_5.10\_2: Schaltung (2) zum Simulationsbeispiel 5.10

\*\*\* Netzliste (2) SB\_5.10 \*\*\*

siehe Lehrbuch – SB\_5.10

Abschn.: **5.10**

DC-Analyse

**AC-Analyse**

Tran.-Analyse

DC-Main-Sweep

DC-Nested-Sweep

**AC-Sweep**

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: **V (outx)**

**OK**



### SB\_5.11: Frequenzgruppenfilter




Bild SB\_5.11\_1: Schaltung des Kombi-Bandpasses im Simulationsbeispiel 5.11

\*\*\* Netzliste (1) SB\_5.11 \*\*\*

```
V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Uq       in  0  AC 1V SIN 0 1V 1kHz 0 0 0
X_N1       0  2  V+ V- out uA741
R_R1       in  1  100k
C_C1       1  2  1.6n
R_RN       2  out 200k
C_CN       2  out 800p
R_Ra       0  out 10MEG
.AC        DEC 1000 10 100k
.TEMP      27
.PROBE
.END
```

Abschn.: **5.10**

DC-Analyse

**AC-Analyse**

Tran.-Analyse

DC-Main-Sweep

DC-Nested-Sweep

**AC-Sweep**

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (out)

OK




Bild SB\_5.11\_2: Schaltung des Doppel-T-Gliedes im Simulationsbeispiel 5.11

\*\*\* Netzliste (2) SB\_5.11 \*\*\*

```
V_Uq      in  0  AC 1V SIN 0 1V 1kHz 0 0 0
R_RT1     in  1  16k
R_RT2     1  out 16k
R_RT3     0  2   8k
C_CT1     in  2  10n
C_CT2     2  out 10n
C_CT3     0  1  20n
R_Ra      0  out 1G
.AC       DEC 1000 10 100k
.TEMP     27
.PROBE
.END
```

- Abschn.: **5.10**
- DC-Analyse
  - AC-Analyse**
  - Tran.-Analyse
  - DC-Main-Sweep
  - DC-Nested-Sweep
  - AC-Sweep**
  - Param.-Sweep
  - Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (out)

OK




Bild SB\_5.11\_3: Gesamtschaltung des Frequenzgruppenfilters im Simulationsbeispiel 5.11

\*\*\* Netzliste (3) SB\_5.11 \*\*\*

```

V_UB1      V+  0  15V
V_UB2      0  V- 15V
V_Uq       in  0  AC 1V SIN 0 1V 1kHz 0 0 0
X_N1       0  GKL  V+ V- GKR  uA741
R_R1       in   4  100k
C_C1       4  GKL  1.6n
R_RN       GKL  3  200k
C_CN       3  GKR  800p
R_RT1      GKL  1  16k
R_RT2      1  GKR  16k
R_RT3      0   2   8k
C_CT1      GKL  2  10n
C_CT2      2  GKR  10n
C_CT3      0   1  20n
R_Ra       0  GKR  10MEG
.AC        DEC  1000  10 100k
.TEMP      27
.PROBE
.END
    
```

Abschn.: **5.10**

DC-Analyse  
**AC-Analyse**  
 Tran.-Analyse

DC-Main-Sweep  
 DC-Nested-Sweep  
**AC-Sweep**  
 Param.-Sweep  
 Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: V (GKR)

OK